
This lecture

1 2 3 4 5 6 7 8 9 10 11 12 13 14 15

I · what a proof is II · how proofs are built III · correct programs IV · AI in the loop

↓ you are here theory in Lean

Equality

 =S y: no
predicate P tells 
and y apart

Eq, Eq.subst

▶ An equality
lets you replace
one side by the
other anywhere.

1

Motives

= is symmetric,
transitive, a
congruence

rw, calc,
congrArg

▶ Each use of an
equality picks a
predicate; rw
picks it for you.

2

Definitional
equality

conversion (βδ)
against proof (=S)

rfl, simp

▶ rfl closes
what computes;
the rest needs a
proof.

3

Uniqueness

∃1, descriptor ι

Classical.choose

▶ A provably
unique thing
may be named,
whatever the
proof.

4

Subsets

ps(S) := S→∗p,
 ϵ V, ⊆

Subtype, Fin

▶ A subset is a
predicate, so
membership is a
proposition.

5

History: what it means to be equal

G. W. Leibniz
1646–1716

Equal means
indiscernible: no

predicate tells the
two apart.

1686 Leibniz: things are the same when one may be
substituted for the other everywhere, salva veri-
tate.

1879 Frege’s Begriffsschrift: the identity sign between
names, with substitution as an axiom; later “the
F”, a definite description.

1908 Zermelo: a subset is cut out of a set by a predi-
cate, the axiom of separation.

1972 Martin-Löf: equality as a type with one construc-
tor and an elimination rule. Lean’s Eq.

Gottlob Frege
1848–1925

ι names the unique
thing with a

property.

Leibniz’s idea becomes =S, Frege’s ι. Photos: Wikimedia. N&G §12.2, §12.7, §13.1

A first piece of mathematics

Lemma. If S is partially ordered by ≤ and has a least element, that element is unique.
Proof. Let m1,m2 be least: m1 ≤m2 and m2 ≤m1, so m1 =m2 by antisymmetry. □

Each numbered gap is filled later; its slide carries the badge. N&G §12.1, Fig. 12.1

What does  = y mean?

Leibniz:  and y are equal when no predicate P tells them apart.

Not an axiom but a definition, so the properties of equality are theorems. N&G §12.2

Equality as a definition
✓ 1

() S : ∗s

(b)  : S

(c) y : S

(1) eq(S, , y) := P:S→∗p. (P⇔ Py) : ∗p

Notation:  =S y for eq(S, , y)

(3) eq-reƒ (S, ) := λP:S→∗p. ⇔ -reƒ (P) :  =S 

• Second order: P ranges over predicates, S→∗p : □.

• Reflexivity is one λ: ⇔ -reƒ (A) := ∧-n(λ:A. , λ:A. ) : A⇔A.

• Even ⇒ in place of⇔ would do.

N&G §12.2, Figs. 12.2–12.3, Rem. 12.2.1–12.2.2

Leibniz equality, written out

-- eq(S,x,y) := P : S → ∗p. (P x ⇔ P y)
-- "x and y are equal iff they are INDISCERNIBLE by any predicate."
def LEq {S : Type} (x y : S) : Prop := ∀ P : S → Prop, P x ↔ P y

-- reflexivity: for every P, ⇔-refl, which Lean calls Iff.rfl
theorem leq_refl {S : Type} (x : S) : LEq x x := fun _ => Iff.rfl

-- ⇒ instead of ⇔ would already do: P x ⇒ P y for all P gives the converse.
def LEq’ {S : Type} (x y : S) : Prop := ∀ P : S → Prop, P x → P y

-- The  ranges over PREDICATES, P : S → ∗p, and S → ∗p : □. So this is
-- a SECOND-ORDER definition, not a first-order ∀ over elements.
#check @LEq -- @LEq : {S : Type} → S → S → Prop

Substitutivity: the rule for using an equality

Substitute equals for equals: if  =S y and P, then Py, for every predicate P.

() S : ∗s

(b) P : S→∗p

(c) , y : S

(d)  :  =S y

(1) 1 := P : P⇔ Py ∀-el on (d)

(2) 2 := ⇔ -e1(P, P y, 1) : P⇒ Py

(e)  : P

(3) eq-sbs(S, P, , y, , ) := 2  : Py ⇒-el

Using an equality means choosing P. N&G §12.2, Fig. 12.4

What Lean uses instead

-- Lean does not use Leibniz’s definition. ‘Eq‘ is an inductive with ONE
-- constructor, and its recursor is what substitutivity becomes.
#print Eq
-- inductive Eq.{u_1} : {α : Sort u_1} → α → α → Prop
-- number of parameters: 2
-- constructors: Eq.refl : ∀ {α : Sort u_1} (a : α), a = a

#check @Eq.subst
-- ∀ {α : Sort u_1} {motive : α → Prop} {a b : α}, a = b → motive a → motive b
-- Compare eq-subs(S,P,x,y,u,v) : P y. Same shape; the predicate P is
-- what Lean calls the ‘motive‘.

The predicate P is what Lean calls the motive. Why Eq.subst is the recursor of Eq: a
later lecture.

The two agree

-- The two notions coincide. One direction substitutes along h,
-- starting from reflexivity . . .
theorem eq_to_leq {S : Type} {x y : S} (h : x = y) : LEq x y :=
fun P => h ▷ Iff.rfl

-- . . . the other instantiates the quantifier at the predicate (x = ·):
-- once more, the proof is the choice of predicate.
theorem leq_to_eq {S : Type} {x y : S} (h : LEq x y) : x = y :=
(h (fun z => x = z)).mp rfl

One direction is substitution with the motive λz. ( = z), and that motive is the whole
proof.

Congruence, proved twice

Congruence. If  :  =S y and ƒ : S→T, then ƒ  =T ƒ y. Two proofs, two choices
of predicate:

first proof

λz:S.Q (ƒ z),
for any Q : T→∗p

Q (ƒ )

assumed

 :  =S y
Q (ƒ y)

close  over Q

second proof

Q1 := λz:S. (ƒ  =T ƒ z) ƒ  =T ƒ 

by refl

 :  =S y
ƒ  =T ƒ y

the goal

The second is one line: eq-cong2 := eq-sbs(S, Q1, , y, , eq-reƒ (T, ƒ )).
N&G §12.3, Figs. 12.5–12.6

Computational Logic in the Era of AI 11 / 44



Choosing Q is choosing the motive

To use  =S y, say which occurrences of  become the hole. That predicate is
Lean’s motive.

N&G §12.3, §12.5

“motive is not type correct”

-- ‘rw‘ builds the MOTIVE Q by abstracting ‘n‘ out of the goal, then checks
-- it. Here the goal mentions ‘v : Fin (n+1)‘ (a natural below n+1, bundled
-- with that proof), whose TYPE mentions n -- so the abstraction is ill-typed:
#guard_msgs(drop error) in
example (n m : Nat) (v : Fin (n + 1)) (h : n = m) : v.val < n + 1 := by

rw [h]
-- Tactic ‘rewrite‘ failed: motive is not type correct:
-- fun _a => ↑v < _a + 1
-- Error: Application type mismatch: The argument v has type
-- Fin (n + 1) but is expected to have type Fin (_a + 1) . . .

-- Two cures: ‘subst‘ replaces n EVERYWHERE, v’s type included; ‘▷‘ takes
-- the motive from the proof term’s type instead of from the goal.
example (n m : Nat) (v : Fin (n + 1)) (h : n = m) : v.val < n + 1 := by

subst h; exact v.isLt
example (n m : Nat) (v : Fin (n + 1)) (h : n = m) : v.val < n + 1 :=

h ▷ v.isLt

No predicate Q with Qn the goal typechecks here: “choose Q” is a real error message.

Symmetry and transitivity, from substitutivity

Two more choices of predicate, with  :  =S y and  : y =S z:

symmetry

Q2 := λz:S. (z =S )  =S 

by refl

 :  =S y
y =S 

the goal

transitivity

Q3 := λ:S. ( =S )  =S y

this is 

 : y =S z
 =S z

the goal

eq-sym(S, , y, ) := eq-sbs(S, Q2, , y, , eq-reƒ (S, )) : y =S 
eq-trns(S, , y, z, , ) := eq-sbs(S, Q3, y, z, , ) :  =S z
Q2 =β  =S : Lean does (con) silently. N&G §12.5, Figs. 12.10–12.12

Computational Logic in the Era of AI 14 / 44



Motives in Lean

-- Congruence: the motive inferred from the goal, or given.
theorem cong’ {S T : Type} (f : S → T) {x y : S} (h : x = y) : f x = f y :=

h ▷ rfl -- motive inferred
theorem cong’’ {S T : Type} (f : S → T) {x y : S} (h : x = y) : f x = f y :=

@Eq.subst S (fun z => f x = f z) x y h rfl -- motive Q1

-- Symmetry and transitivity, with Q2 and Q3 given.
theorem sym’ {S : Type} {x y : S} (h : x = y) : y = x :=

@Eq.subst S (fun z => z = x) x y h rfl -- Q2 x is x = x
theorem trans’ {S : Type} {x y z : S} (h : x = y) (k : y = z) : x = z :=

@Eq.subst S (fun w => x = w) y z k h -- Q3 y is x = y

-- In the library: ‘congrArg‘, ‘Eq.symm‘, ‘Eq.trans‘.

rw chooses the motive for you

-- ‘rw [h]‘ is eq-subs with the predicate chosen FOR you: it abstracts every
-- occurrence of h’s left side out of the goal, and that abstraction is Q.
theorem rw1 (a b : Nat) (h : a = b) (f : Nat → Nat) : f a + a = f b + b := by

rw [h] -- Q := fun z => f z + z = f b + b; the rest is Eq.refl

example (a b : Nat) (h : a = b) (k : b = 0) : a = 0 := by
rw [h, k] -- one equation after another, each left to right

example (a b : Nat) (h : b = a) : a = b := by
rw [← h] -- right to left

example (a b : Nat) (h : a = b) (k : a + 1 = 5) : b + 1 = 5 := by
rw [h] at k; exact k -- in a hypothesis

#print rw1
-- fun a b h f => Eq.mpr (id (congrArg (fun _a => f _a + _a = f b + b) h))
-- (Eq.refl (f b + b))

Read the term: congrArg at the motive is congruence, Eq.mpr is substitution along it,
and Eq.refl closes what is left. All of the equality theory so far is inside one rw.

Chains of steps: calc

-- ‘calc‘ chains equalities step by step. Each step names its
-- justification; transitivity is applied for you.
def d (n : Nat) : Nat := n + n

example (a b : Nat) (h : a = b) (k : b = 5) : d a = 10 :=
calc d a = d b := by rw [h] -- congruence

_ = d 5 := by rw [k]
_ = 10 := rfl -- definitional: δ, then arithmetic

-- Mixed relations chain too; the result is the strongest one used.
example (a b c e : Nat) (h1 : a ≤ b) (h2 : b < c) (h3 : c ≤ e) : a < e :=

calc a ≤ b := h1
_ < c := h2
_ ≤ e := h3

Orders
✓ 3

“Partial order” is five definitions in one context:

() S : ∗s

(b) ≤ : S→S→∗p

Notation:  ≤S y for ≤ y

(1) reƒ (S,≤) := ∀:S. ( ≤S ) : ∗p

(2) trns(S,≤) := ∀, y, z:S. ( ≤S y⇒ y ≤S z⇒  ≤S z) : ∗p

(3) pre-ord(S,≤) := reƒ (S,≤)∧ trns(S,≤) : ∗p

(4) ntsymm(S,≤) := ∀, y:S. ( ≤S y⇒ y ≤S ⇒  =S y) : ∗p

(5) prt-ord(S,≤) := pre-ord(S,≤)∧ ntsymm(S,≤) : ∗p

3 Lest(S,≤,m) := ∀n:S. (m ≤S n). N&G §12.4, Fig. 12.7, Rem. 12.4.1

A skeleton proof: the types, and nothing else

First the flags and the types of the lines; every proof is still a hole.

Filling the holes fills 2 . N&G §12.4, Fig. 12.8

The proof objects, filled in
✓ 2

() S : ∗s | ≤ : S→S→∗p | r : prt-ord(S,≤)

(d) m1,m2 : S

(e)  : ∀n:S. (m1 ≤S n) |  : ∀n:S. (m2 ≤S n)

(1) 1 := m2 : m1 ≤S m2 ∀-el

(2) 2 := m1 : m2 ≤S m1 ∀-el

(4) 4 := ∧-e2(. . . , r) : ntsymm(S,≤) ∧-el2, then (con)

(6) 6 := 4m1m2 : m1 ≤S m2 ⇒m2 ≤S m1 ⇒m1 =S m2 ∀-el twice

(8) 8 := 6 1 2 : m1 =S m2 ⇒-el twice

(9) 9(S,≤, r) := λm1m2:S. λ. λ. 8 : ∀m1,m2:S. (· · · ⇒m1 =S m2) (bst)

2 t3 is 8. Lines (3), (5), (7) only unfold, by (con). N&G §12.4, Fig. 12.9

The same proof, as named lines

-- The flag derivation, line for line. A ‘have‘ is a named line; ‘exact‘
-- is the last.
theorem least_unique {S : Type} (le : S → S → Prop) (r : PartOrd le)

(m1 m2 : S) (u : ∀ n, le m1 n) (v : ∀ n, le m2 n) : m1 = m2 := by
have a1 : le m1 m2 := u m2 -- (1) ∀-el on u
have a2 : le m2 m1 := v m1 -- (2) ∀-el on v
have a4 : Antisymm le := r.2 -- (3)-(4) ∧-el2 on r
have a6 : le m1 m2 → le m2 m1 → m1 = m2 := a4 m1 m2 -- (5)-(6)
exact a6 a1 a2 -- (7)-(8)

#print least_unique
-- fun {S} le r m1 m2 u v => have a1 := u m2; have a2 := v m1;
-- have a4 := r.right; have a6 := a4 m1 m2; a6 a1 a2

The haves survive in the term as local definitions: the flag derivation, folded into one
object.

Two equalities

Teal sits inside gold, not the other way round: that is why rw exists. N&G §4.7,
§9.7, §12.2

Definitional and propositional, side by side

definitional propositional

what it is a judgement a proposition

written M=βδ N M =S N

comes from the (con) rule a definition

decided by the kernel nobody: it needs a proof

closed by rfl rw, Eq.subst, calc

n + 0 = n is definitional: addition recurses on its second argument. 0 + n = n is
not: it needs an induction. N&G §4.7, §12.2

Computational Logic in the Era of AI 23 / 44



Where rfl stops, and what takes over

-- ‘Nat.add‘ recurses on its SECOND argument, so one of these computes and
-- the other is stuck on the variable n:
example (n : Nat) : n + 0 = n := rfl
#guard_msgs(drop error) in
example (n : Nat) : 0 + n = n := rfl
-- The type of ‘Eq.refl ?a‘ . . . is not definitionally equal to ‘0 + n = n‘

-- What closes it is a theorem from Δ -- proved by induction, next lecture:
example (n : Nat) : 0 + n = n := Nat.zero_add n

-- With n closed, computation is back, and ‘decide‘ runs it in the kernel:
example : 0 + 5 = 5 := by decide
Past the boundary: a lemma, a closed instance the kernel runs, or rewriting to a normal
form.

Rewriting to a normal form

A rewrite rule is an equation read one way, s→ t. simp rewrites with every
lemma tagged @[simp]: its simp set.

N&G §1.9

Unique existence
✓ 4 5

4 Three grades of “exists”, for P : S→∗p:

definition reads e.g.

∃≥1(S, P) ∃:S. P at least one

∃≤1(S, P) ∀y, z:S. (Py⇒ P z⇒ y =S z) at most one

∃1(S, P) ∃≥1(S, P)∧∃≤1(S, P) exactly one

an element with P

5 Line (9) of the proof is ∃≤1 for P := Lest; with  : ∃≥1, ∧-in gives ∃1.

Theorems and proofs
If  ⊢ (̄) := N : ∗p and  ⊢ b(̄) := M : N, then (̄) is a theorem and b(̄) its
proof. N&G §12.6, Figs. 12.14–12.15

The descriptor ι

What is unique can be named: ι:S(P), the  with P. Two primitive
definitions, the object and its property:

() S : ∗s | P : S→∗p |  : ∃1:S. P

(1) ι(S, P, ) := ⊥⊥ : S no body

Notation: ι:S(P) for ι(S, P, )

(2) ι-prop(S, P, ) := ⊥⊥ : P (ι:S(P)) no body

• Whatever has P is the ι: ∀z:S. (P z⇒ z =S ι:S(P)).

• So the proof  does not matter: ι(S, P, 1) =S ι(S, P, 2).
• The minimum: Mn := ι(S, λm. Lest(S,≤,m), . . . ).

N&G §12.7, Figs. 12.16–12.18, Rem. 12.7.2

Classical.choose is ι

-- ∃1 is "at least one" and "at most one". Core Lean has no ∃1, so the
-- definition is written out.
def ExUnique {S : Type} (P : S → Prop) : Prop :=

(∃ x, P x) ∧ ∀ y z, P y → P z → y = z

-- ι(S,P,u) := ⊥⊥ : S and ι-prop(S,P,u) := ⊥⊥ : P (ι . . .).
-- Lean’s are ‘Classical.choose‘ and ‘choose_spec‘, built on an axiom:
noncomputable def iota {S : Type} (P : S → Prop) (u : ExUnique P) : S :=

Classical.choose u.1
theorem iota_prop {S : Type} (P : S → Prop) (u : ExUnique P) : P (iota P u) :=

Classical.choose_spec u.1
#print axioms iota_prop -- ’iota_prop’ depends on axioms: [Classical.choice]

-- Whatever satisfies P is the ι.
theorem eq_iota {S : Type} (P : S → Prop) (u : ExUnique P) (z : S) (v : P z) :

z = iota P u := u.2 z _ v (iota_prop P u)

noncomputable: a primitive constant has no body to run, and the axiom audit names it.

Irrelevance of proof

-- λD does not make proof irrelevance a principle. Lean’s kernel has it
-- for Prop, definitionally: two proofs of one proposition are rfl-equal.
example (p : Prop) (h1 h2 : p) : h1 = h2 := rfl

-- So the worry that ι(S,P,u1) and ι(S,P,u2) might differ
-- cannot even be stated as a difference:
example {S : Type} (P : S → Prop) (u1 u2 : ExUnique P) :

iota P u1 = iota P u2 := rfl

-- What keeps it consistent: a proof may not be turned into DATA. A Prop’s
-- recursor eliminates only into Prop, so no function can tell two
-- proofs apart:
#guard_msgs(drop error) in
def theWitness (h : ∃ n : Nat, 3 < n) : Nat := match h with | 〈n, _〉 => n
-- error: recursor ‘Exists.casesOn‘ can only eliminate into ‘Prop‘

λD proves the equation; Lean’s kernel makes it a conversion. That is one of the places
Lean is not λD.

Why subsets cannot be types

Suppose a subset T ⊆ S were a type, so that c : T.

one term, two types

S

T
c

c : S and c : T:
Uniqueness of Types fails

type checking becomes proof search

type-check FM, with
F the square root

needs M : { ∈ R |  ≥ 0}

find a proof of M ≥ 0

undecidable in general
(Church, Turing 1936)

So typing stays decidable, and subsets need another representation. N&G §13.1

Subsets are predicates

A subset of S is a predicate on S; predicates are already in λD.

() S : ∗s

(1) ps(S) := S→∗p : □

(b)  : S | V : ps(S)

(2) eement(S, , V) := V  : ∗p Notation:  ϵ V

(c) V,W : ps(S)

(3) ⊆ (S,V,W) := ∀:S. ( ϵ V ⇒  ϵW) : ∗p

(4) ∪(S,V,W) := λ:S. ( ϵ V ∨  ϵW) : ps(S)

 ϵ V is a proposition to prove, not a typing  : V. N&G §13.2, Figs. 13.1–13.2

ps(S) in Lean

-- Lean core has no ‘Set‘; these lines are all of it.
abbrev PS (S : Type) : Type := S → Prop -- (1) ps(S)
def Mem {S : Type} (V : PS S) (x : S) : Prop := V x -- (2) x ϵ V
instance {S : Type} : Membership S (PS S) := 〈Mem〉 -- write x ∈ V
def Subs {S : Type} (V W : PS S) : Prop := -- (3) V ⊆ W
∀ x : S, x ∈ V → x ∈ W

infix:50 " ⊑ " => Subs
def IS {S : Type} (V W : PS S) : Prop := V ⊑ W ∧ W ⊑ V -- same elements
def diff {S : Type} (V W : PS S) : PS S := fun x => x ∈ V ∧ ¬ x ∈ W -- V \ W
def compl {S : Type} (V : PS S) : PS S := fun x => ¬ x ∈ V -- V^c

example (S : Type) (V W : PS S) :
(V ⊑ W) = ∀ x : S, x ∈ V → x ∈ W := rfl -- the guarded ∀, verbatim

The instance line tells Lean that  ∈ V means Mem V x. Later slides write V ⊆ W as
V ⊑W.

Computational Logic in the Era of AI 32 / 44



Membership needs a proof

-- x ∈ V is not a typing x : V but a PROPOSITION. To establish it we
-- have to give a PROOF.
def Evens : PS Nat := fun n => n % 2 = 0

-- ‘4 ∈ Evens‘ unfolds to ‘4 % 2 = 0‘, so a computation proves it.
example : 4 ∈ Evens := show 4 % 2 = 0 by decide
#guard_msgs(drop error) in
example : 5 ∈ Evens := show 5 % 2 = 0 by decide -- fails: 5 % 2 = 1

-- ‘n ∈ Evens‘ IS ‘n % 2 = 0‘, up to unfolding: a proof of one is a
-- proof of the other, with nothing to convert.
example (n : Nat) (h : n % 2 = 0) : n ∈ Evens := h

Every quantifier ranges over a type

V is not a type, so ∀ : V is not well formed. Quantify over S and guard:

N&G §13.2, Ex. 13.2.1

Guarded quantifiers

-- ∀x ∈ Evens. x·x is even: quantify over Nat, guard with ⇒.
example : ∀ x : Nat, x ∈ Evens → (x * x) % 2 = 0 := by
intro x (hx : x % 2 = 0) -- any x, assuming the guard
simp [Nat.mul_mod, hx]

-- ∃x ∈ Evens. x > 3: guard with ∧; the witness 4 meets both halves.
example : ∃ x : Nat, x ∈ Evens ∧ x > 3 :=
〈4, show 4 % 2 = 0 by decide, by decide〉

-- With ⇒ instead, x = 1 would do: 1 /∈ Evens makes the ⇒ vacuous.
example : ∃ x : Nat, x ∈ Evens → x > 1000 :=
〈1, fun h => absurd (show 1 % 2 = 0 from h) (by decide)〉

A proof about sets

With V \W := λ:S. ( ϵ V ∧¬( ϵW)) and Wc := λ:S.¬( ϵW): if V ⊆ Wc, then
V \W = V.

() S : ∗s | V,W : ps(S)

(b)  : S |  :  ϵ V \W

(2) 2 := . . . use ∧-el1 on  . . . :  ϵ V  :  ϵ V ∧¬( ϵW), by (con)

(4) 4(S,V,W) := . . . use ⇒-in and ∀-in . . . : V \W ⊆ V

(c)  : V ⊆ Wc

(d)  : S |  :  ϵ V

(5) 5 :=  :  ϵWc ∀-el, ⇒-el

(8) 8 := . . . use ∧-in on  and 5 . . . :  ϵ V \W  ϵWc =β ¬( ϵW)

(10) 10(S,V,W,) := . . . : V ⊆ V \W

(12) 12(S,V,W) := . . . use ∧-in on 4 and 10, then ⇒-in . . . : (V ⊆ Wc)⇒ (V \W = V)

Each “use . . . ” names one rule. Only 10 depends on . N&G §13.2, Fig. 13.4

The same proof, as a script

-- The flag derivation, as a script. ‘intro x v‘ raises the flags x : S
-- and v; the hints "use ∧-el1", "use ∧-in", "use ⇒-in and ∀-in" become
-- ‘.1‘, ‘〈_, _〉‘ and ‘intro‘.
theorem diff_eq {S : Type} (V W : PS S) (u : V ⊑ compl W) :

IS (diff V W) V := by
constructor
· intro x v -- lines (1)-(4): no use of u
exact v.1 -- (2) ∧-el1 on v : x ϵ V ∧ ¬(x ϵ W)

· intro x v -- lines (5)-(10)
have a5 : x ∈ compl W := u x v -- (5) u x v : x ϵ W^c
exact 〈v, a5〉 -- (7)-(8) ∧-in: x ϵ V\W

-- The second half needs u and the first does not, and the parameter
-- lists show it (a4(S,V,W) against a10(S,V,W,u)).

A specification is a predicate, that is, a subset

a specification, in words as a predicate

the sorted lists λs:LstN. sorted(s) : ps(LstN)
the inputs on which ƒ returns 0 λn:N. (ƒ n =S 0) : ps(N)
the states from which a loop terminates λs:Stte. . . . : ps(Stte)
an assertion about a program state λs:Stte. (0 < s) : ps(Stte)
“P entails Q”, P the stronger P ⊆ Q
a predicate transformer a map ps(Stte)→ps(Stte)

“The program satisfies it” is a membership  ϵ V, to be proved. Nothing is
added to the type theory to state a specification.

N&G §13.2

The other route: carry the proof

A subtype value is the element with its proof stamped on.

The proof travels with the element: checked, never searched for.

Subtypes in Lean

-- Bundle the element WITH its proof:
-- the proof is carried, not searched for.
def NonNeg := {n : Int // 0 ≤ n}

def three : NonNeg := 〈3, by decide〉
#eval three.val -- 3
#check three.property
-- three.property : 0 ≤ three.val

-- The cost: ‘3 : Int‘ and ‘three : NonNeg‘
-- are different terms of different types;
-- a coercion moves between them.
-- ‘Fin n‘ is ‘{i : Nat // i < n}‘.

Extensionality is an axiom

Two subsets with the same elements need not be equal: = compares the
terms, and no rule looks inside a function. So it is added as an axiom.
-- Same elements, different terms: 0 + n does not compute to n.
def A : PS Nat := fun n => 0 + n = 5
def B : PS Nat := fun n => n = 5
theorem same_elements : ∀ x, x ∈ A ↔ x ∈ B := by
intro x; show 0 + x = 5 ↔ x = 5; rw [Nat.zero_add]

#guard_msgs(drop error) in
example : A = B := rfl -- fails: equality compares the terms

-- The axioms: functions equal pointwise, propositions equal if ↔.
theorem setext {α : Type} {V W : PS α} (h : ∀ x, x ∈ V ↔ x ∈ W) : V = W :=

funext (fun x => propext (h x))
example : A = B := setext same_elements

#print axioms setext -- [propext, Quot.sound]: funext comes from quotients

N&G §13.2, Fig. 13.5

What is still missing: numbers

Equality, orders and sets are definitions. Numbers are not: in λD they are six
primitive definitions, and they do not compute.

The Peano axioms in λD

(1)–(3) N := ⊥⊥ : ∗s, 0 := ⊥⊥ : N, s := ⊥⊥ : N→N the set, zero, successor

(4) -nt1 := ⊥⊥ : ∀:N.¬(s =N 0)
(5) -nt2 := ⊥⊥ : ∀, y:N. (s =N s y⇒  =N y)
(6) -nt3(P) := ⊥⊥ : (P0∧ ∀:N. (P⇒ P (s)))⇒ ∀:N. P  induction

And without recursive definitions, ƒc is not a program: it cannot be executed,
so ƒc(3) = 6 requires a proof.

Next lecture
Inductive types: two constructors in place of six axioms, and ƒc3 = 6 by rfl.
N&G §8.2, §14.1

